
Introduction

Mouse events traditionally provide the user with positional information but not velocity or acceleration. Presumably, this kind of information would be useful to gestural systems or to applications that can take advantage of the additional time-based information.



In the last year, we spent considerable time reading long documents and we grew tired of manually scrolling the text. What we needed was a simple auto-scrolling facility that would eliminate the requirement to physically drag a scroll bar or click on a scrolling button. By dragging the scroll bar in the normal manner, we wanted to get the standard behavior but if we gave it a little throw, we hoped the scrolling would continue on its own.



Designing The Application

An auto-scrolling application is as simple as a window with a text pane that can auto-scroll. Smalltalk Express, for example, has a simple text window that can be created and opened as follows:



	TextWindow new 

		openOn: 'Testing...'.

	

In our case, we will use a simple variant that is created and opened in the same way.



    SpeedReadingWindow new 

		openOn: 'Testing...'.

	

The only difference between a text window and a speed-reading window is that the latter makes use of a speed-reading text pane instead of a standard text pane. The SpeedReadingWindow class hides the details of the text pane replacement.



Actually, the above example creates a text window with one line of text—not enough to illustrate the auto-scrolling capability. So the first thing that we need to do is click on Open in the File menu to open another speed-reading window on a large file (a file prompter appears). Somewhat surprisingly, we can get a file prompter to appear and get a speed-reading window on that file by executing the following. It’s not necessary to open the window that initiates the file prompter.



    SpeedReadingWindow new 

		openFile



The SpeedReadingTextPane class will have to deal with a new way of interpreting scroll bar manipulations. We will need to record not only positional information but also time and enough prior information that we will be able to determine the velocity of the scroll bar dragging operation. Rather than store all this information directly in the speed-reading text pane, we might as well store it in a more appropriate object—a speed-reading mouse. So we introduced a class SpeedMouse for this purpose.



The Speed-Reading Window



The speed-reading window class can be created as a subclass of TextWindow with all references to TextPane replaced by SpeedReadingTextPane. But we didn’t have to be that baroque. As you can see in Listing 1, only 3 methods were overridden. Class method textWindowClass and instance method textPaneClass had to be changed to return SpeedReadingTextWindow and SpeedReadingTextPane respectively. If that’s all we added to the subclass, however, openFile would create a normal text window rather than another speed-reading text window. The problem is that openFile is implemented as follows:



	openFile

		 "The user selected Open... from the File menu."

		self mainView openFile



The main view is an instance of the TopPane class which contains an implementation of openFile that creates a text window. The proper solution requires the introduction of a new kind of top pane that would have to be overridden with a simple one line method, textWindowClass. The alternative (which eliminates the need to create yet another class) was to copy the openFile method into our SpeedReadingTextWindow class.



The Speed-Reading Text Pane

When a scroll bar is manipulated in a pane, Windows 95 generates either a WM_VSCROLL or a WM_HSCROLL event (depending on the type of scroll bar). In our case, we are only interested in vertical scroll events which results in the execution of a method called wmVScroll:with: (see Listing 2). An equivalent facility is also provided in VisualAge. 



Two pieces of information are provided by the event: the type of user manipulation and the position of the scroll bar. Type information is one of a number of constants—SbThumbtrack when the scroll bar is dragged, SbThumbposition when dragging stops, SbEndscroll when the mouse is released, SbLineup when the up arrow is pushed, SbLinedown when the down arrow is pushed, SbPageup when the page up key is pressed, SbPagedown when the page down key is pressed, …



In our case, we are really interested in recording the new position in the speed mouse when SbThumbtrack and SbThumbposition are signaled. When SbEndscroll is signaled, we want to analyze the speed of the mouse when we released it (afterwards, its velocity is set to zero). If it was moving fast, we would like to auto-scroll reasonably fast. If it was moving slowly, we would like to scroll slowly. Otherwise, we would not auto-scroll at all.



Our first implementation (not shown) had simply gone into a loop scrolling the text according to the results of our query: #fast, #slow, #notAtAll. But that placed the application in a busy loop and also prevented it from being interruptible. So what we did next was create a timer that triggers the autoScroll method. For fast scrolling, the timer is triggered at a faster rate. 



The timer is created in method activateAutoScroll: and destroyed in deactivateAutoScroll:. You will also notice that if the timer is already active (i.e., autoScrolling returns true) when the wmVScroll:with: method executes, we deactivate it (thereby stopping the on-going auto-scrolling).



If the text pane class that our new class inherits from supported smooth scrolling, we would have been content with two instance variables: “speedMouse” and “timer.” Unfortunately, the minimum scroll amount that it supported was the height of one character; e.g., 16 pixels for a 16 point font size; it wasn’t very smooth. We had to simulate the smooth scrolling by physically moving the pixels up or down as appropriate by 1 pixel. So we introduced the additional variable “pixelsScrolled” for counting up between 1 and the font size. Variable “pixelsScrolled” is used in method autoScroll.



The Speed Mouse

The velocity sensitive mouse was designed to keep track of a position, velocity, and acceleration although we aren’t making any use of the latter. To do that we had to keep track of the time that the position was set. Since velocity is a change of position, we had to keep both an old position (with its time) and a new position (also with its time). The same applied to velocity since acceleration is a change of velocity. 



Our initial goal had been to say something simple like “positionNow: aPoint” (see Listing 3) and have the speed mouse maintain and revise the velocity and acceleration. We tried a lot of schemes that did not work well and we can’t remember them all now but a few do come to mind. In one approach, we just discarded the oldest position (and time) and tried to compute a new velocity and acceleration. Most of the time, the change in time was zero since the microsecond clock wasn’t accurate enough to provide distinctions between, say, two successive SbThumbtrack events or between an SbThumbtrack and an SbThumbposition event. Another implementation maintained a history of positions in an ordered collection permitting us to analyze the changes over a more significant time period. We had sufficient information to track all changes in position. We could also deduce when the velocity was increasing, decreasing, reaching zero, changing direction, … But in the end, it was just too complicated to deal with all the data. 



We finally settled for an approach where the old time, old position, and old velocity was kept until either a significant amount of time elapsed or a significant change in position occurred (even in a small amount of time). Until a revision was required (determined by method revisionRequired), we would change only the current position, velocity, and acceleration. When a revision was allowed, the old values were replaced (pushed aside) by the new ones.



Initially, the break point values that indicated a significant change were class variables but we finally decided to make it user specifiable. So we introduced two more instance variables: “significantDeltaTime” and “significantDeltaPosition.”

 

Conclusions

Initially, we wanted to experiment with a velocity sensitive mouse and illustrate its use in an application that permitted a user to grab and throw a scroll bar to start an automatic scrolling facility. In hindsight, the mouse seems to be a relatively minor part of the experiment. 



Although the experiment works, it doesn’t work particularly well. To get fast smooth scrolling requires a timer period that is too fast. I suspect that it is currently less than the minimum timer resolution of the clock. As a consequence, there is little difference between fast and slow scrolling. We could deal with this by extending the existing code to scroll by several pixels at a time but we didn’t have the time to make that extension.



Even without the notion of a velocity sensitive mouse, it would be useful for word processors to support auto-scrolling that can be geared to a person’s reading speed.





Listing 1 The speed-window class.



class:								SpeedReadingWindow

superclass:						TextWindow 

instance variables:		speedMouse 



class methods

examples

example1

	"SpeedReadingWindow example1 openFile"

	^self new openOn: 'Speed window test...'.



overrides

textWindowClass

	^self



instance methods

overrides

textPaneClass

	^SpeedReadingTextPane

   

openFile

	"Private - User selected Open... from the File menu."

	| dialog file |

	(dialog := FileDialog new openFile ) isNil

		ifTrue: [^self ].

	(file := dialog file) isNil

		ifTrue: [ ^self ].

	(File exists: file) ifFalse: [

		(MessageBox confirm: 'File does not exist.  Create it?' )

			ifTrue: [File createFileNamed: file]

			ifFalse: [^self]].

	file := File pathNameReadOnly: file.

	self class textWindowClass new openOnFile: file.

	file close







Listing 2 The speed-reading text class.



class:								SpeedReadingTextPane

superclass:						TextPane 

instance variables:		speedMouse timer pixelsScrolled

pool dictionaries: 		OperatingSystemConstants 



class methods

constants

fastScrollPeriod

	^100

slowScrollPeriod

	^400



querying

periodForSpeed: speed fontSize: pixels

	^((self periodForSpeed: speed) / pixels) rounded



periodForSpeed: speed

	speed abs > 0.02 ifTrue: [^self fastScrollPeriod].

	speed abs > 0.01 ifTrue: [^self slowScrollPeriod].

	^0



instance methods

initializing

initialize

	super initialize.

	speedMouse := SpeedMouse new

		significantDeltaPosition: 10;

		significantDeltaTime: 40.

	autoScrolling := false



get/set

speedMouse

	^speedMouse

speedMouse: aSpeedMouse

	speedMouse := aSpeedMouse



querying

instantaneousSpeed

	^self speedMouse velocity asPoint y

speed

	^self speedMouse oldVelocity asPoint y



opening/closing

close

	self deactivateAutoScroll.

	^super close



overrides

wmVScroll: wordInteger with: longInteger

	"Private - Process the vertical scroll message."

	| type position period |

	type := wordInteger lowWord. 

	position := wordInteger highWord.

	self autoScrolling ifTrue: [self deactivateAutoScroll. ^nil].

	type = SbThumbposition ifTrue: ["end of tracking"

		 self speedMouse positionNow: position].

	type = SbThumbtrack ifTrue: [

		self speedMouse positionNow: position].

	type = SbEndscroll ifTrue: [

		period := self class

			periodForSpeed: self instantaneousSpeed

			fontSize: self font height.

		period > 0 ifTrue: [

			self speedMouse velocityNow: 0.0.

			self activateAutoScroll: period]].

	^super wmVScroll: wordInteger with: longInteger



auto scrolling

autoScroll

	| speed pen |

	self autoScrolling ifFalse: [^self].

	pixelsScrolled := pixelsScrolled + self speed sign negated.

	pixelsScrolled abs = self font height

		ifTrue: [

			topCorner := topCorner + 

				(0@pixelsScrolled sign negated).

			self display; updateVerticalSlider.

			(topCorner y <= 1 or: [

			topCorner y = textHolder extent y]) ifTrue: [

				self deactivateAutoScroll].

			pixelsScrolled := 0]

		ifFalse: [

			pen := self pen.

			pen selfCopyToX: 0 Y: pixelsScrolled sign.

			pixelsScrolled positive

				ifTrue: [

					pen blank: (0@0 extent: pen width@1)]

				ifFalse: [

					pen blank: (0@pen height-1 

						extent: pen width@1)]]



activateAutoScroll: period

	"Remove old timer."

	self deactivateAutoScroll.

	"Can't go up if at the top."

	(topCorner y = 0 and: [self speed negative]) ifTrue: [^self].

	"Can't go down if at the bottom."

	(topCorner y = textHolder extent y and: [

	self speed positive]) ifTrue: [^self].

	"Set up new timer."

	timer := Timer new

		setName: 'auto scroll timer';

		when: #ticked send: #autoScroll to: self;

		period: period;

		start.

	"Prepare for slow scrolling."

	pixelsScrolled := 0

 

deactivateAutoScroll

	timer isNil ifTrue: [^self].

	timer stop.

	timer := nil



autoScrolling

	^timer notNil







Listing 3 The velocity-sensity mouse class.



class:								SpeedMouse

superclass:						Object 

instance variables:		time oldTime position oldPosition 

										velocity oldVelocity 

										acceleration significantDeltaPosition 

										significantDeltaTime



class methods

creating

new

	^super new initialize



instance methods

initializing

initialize

	| zero now |

	zero := 0.0.

	now := Time millisecondClockValue.

	self oldTime: now; time: now.

	self oldPosition: zero; position: zero.

	self oldVelocity: zero; velocity: zero.

	self acceleration: zero.

	self significantDeltaPosition: 10@10.

	self significantDeltaTime: 40.



get/set

time

	^time

time: aTime

	time := aTime

oldTime

	^oldTime

oldTime: aTime

	oldTime := aTime



position

	^position

position: aPoint

	position := aPoint

oldPosition

	^oldPosition

oldPosition: aPoint

	oldPosition := aPoint

 

velocity

	^velocity

velocity: aPoint

	velocity := aPoint

oldVelocity

	^oldVelocity

oldVelocity: aPoint

	oldVelocity := aPoint

 

acceleration

	^acceleration

acceleration: aPoint

	acceleration := aPoint



significantDeltaTime

	^significantDeltaTime

significantDeltaTime: aFloat

	significantDeltaTime := aFloat

   

significantDeltaPosition

	^significantDeltaPosition

significantDeltaPosition: aPoint

	significantDeltaPosition := aPoint



querying

deltaTime

	^self time - self oldTime

deltaPosition

	^self position - self oldPosition

deltaVelocity

	^self velocity - self oldVelocity



tracking

positionNow: aPoint

	| now |

	now := Time millisecondClockValue.

	self updateFromPosition: aPoint time: now.

	self revisionRequired ifTrue: [

		self pushTime; pushPosition; pushVelocity]



velocityNow: aPoint

	self pushVelocity; velocity: aPoint



private/tracking

updateFromPosition: aPoint time: aTime

	"Update position, velocity, acceleration; the latter two, 

	only if the time delta is not zero."

	| duration |

	self time: aTime; position: aPoint.

	duration := self deltaTime asFloat.

	duration < 0.5 ifTrue: [^self].

	self

		velocity: self deltaPosition / duration;

		acceleration: self deltaVelocity / duration



revisionRequired

	| delta |

	delta := self deltaTime.

	delta >= self significantDeltaTime ifTrue: [^true].

	delta == 0 ifTrue: [^false].

	^self deltaPosition abs >= self significantDeltaPosition



private/pushing

pushTime

	self oldTime: self time

pushVelocity

	self oldVelocity: self velocity

pushPosition

	self oldPosition: self position

